
Ajout bibliothèque, Open Query File 
 
 
 
CL : RCLPRT                
  100 /* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * */                                  
  200 /* Propriétaire . . . . . . . . . :  GB                                      */                                  
  300 /* Nom du programme . . . . . . . :  CL1                                     */                                  
  400 /* Nom de la bibliothèque . . . . :  TEST                                    */                                  
  500 /* Fichier source initial . . . . :  QCLPSRC                                 */                                  
  600 /* Compiler avec  . . . . . . . . :  OWNER                                   */                                  
  700 /*                                                                           */                                  
  800 /* Texte  . . : exemple                                                      */                                  
  900 /*                                                                           */                                  
 1000 /* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * */                                  
 1100              PGM        PARM(&SOC &ETAB &DATE &DATF &INRE &USER)                                                 
 1200                                                                                                                  
 1300              DCL        VAR(&SOC)     TYPE(*CHAR) LEN(2)                                                         
 1400              DCL        VAR(&ETAB)    TYPE(*CHAR) LEN(2)                                                        
 1500              DCL        VAR(&DATE)    TYPE(*CHAR) LEN(8)                                                         
 1600              DCL        VAR(&DATF)    TYPE(*CHAR) LEN(8)                                                         
 1700              DCL        VAR(&INRE)    TYPE(*CHAR) LEN(3)                                                         
 1800              DCL        VAR(&MCUM)    TYPE(*CHAR) LEN(6)                                                         
 1900              DCL        VAR(&MCUF)    TYPE(*CHAR) LEN(6)                                                         
 2000              DCL        VAR(&USER)    TYPE(*CHAR) LEN(10)                                                        
 2100              DCL        VAR(&BIB)     TYPE(*CHAR) LEN(10)                                                        
 2200                                                                                                                  
 2300 /* -------------------------------------------------------------------------- */                                 
 2400 /* Standard Gestion des erreurs */                                                                               
 2500  DCLMSG:                                                                                                         
 2600              DCL  &MSGF    *CHAR  10                                                                             
 2700              DCL  &MSGFLIB *CHAR  10                                                                             
 2800              DCL  &MSGID   *CHAR   7                                                                            
 2900              DCL  &MSGDTA  *CHAR  80                                                                            
 3000              DCL  &RTNTYPE *CHAR   2          /* Type de retour message */                                      
 3100              DCL  &DIAG    *CHAR   2  '02'                                                                      
 3200              DCL  &EXCP    *CHAR   2  '15'                                                                      
 3300              DCL  &CANCELD *CHAR  30                                                                             
 3400 /* Moniteur général d'erreurs et routine de gestion des erreurs  */                                              
 3500              MONMSG     MSGID(CPF0000 MCH0000) EXEC(GOTO +                                                       
 3600                           CMDLBL(ERREUR))                                                                        
 3700 /* -------------------------------------------------------------------------- */                                 
 3800 /* Recherche et ajout d'une bib à la liste */                                     
 3900              CHGVAR     VAR(&BIB)  VALUE(%SST(*LDA 108 10))                                                      
 4000              ADDLIBLE   LIB(&BIB) POSITION(*LAST)                                                               
 4100              MONMSG     MSGID(CPF2103 CPD0078 CPF0001)                                                           
 4200              MONMSG     MSGID(CPF2106 CPF2110 CPF2113 CPF2118 +                                                  
 4300                           CPF2176 CPF2182 CPF9807) EXEC(DO)                                                      
 4400              DSPJOBLOG  OUTPUT(*PRINT)                                                                          
 4500              GOTO       CMDLBL(£ENDPGM)                                                                          
 4600              ENDDO                                                                                               
 4700 /* --------- */                                                                                                 
 4800 /* Vérification si le fichier de travail est utilisé, ensuite clear des données */ 
 4900              ALCOBJ     OBJ((fileWork1 *FILE *EXCL *FIRST)) +                                                     
 5000                           WAIT(10)                                                                              
 5100              MONMSG     MSGID(CPF1002 CPF9898) EXEC(DO)                                                          
 5200              SNDPGMMSG  MSG('Programme en cours d''utilisation pour +                                           
 5300                           le moment') MSGTYPE(*INFO)                                                             
 5400              GOTO       £ENDPGM                                                                                 
 5500              ENDDO                                                                                               
 5600              DLCOBJ     OBJ((fileWork1 *FILE *EXCL *FIRST))                                                       
 5700              CLRPFM     FILE(fileWork1)                                                                          
 5800 /* Changement de la priorité d'exécution du job */                
      /* non affichage des messages */                
 5900              CHGJOB     RUNPTY(50) STSMSG(*NONE)                                                                
7100 /* --------- */                                                                                              
 7200 /* Sélection des enregistrements pour … */                                                                                               
 7300              CHGVAR     VAR(&MCUM) VALUE(%SST(&DATE 1 6))                                                    
 7400              CHGVAR     VAR(&MCUF) VALUE(%SST(&DATF 1 6))                                                    
 7500              OVRDBF     FILE(FILE2) MBR(*FIRST) SHARE(*YES)                                                  

 7600              OPNQRYF    FILE((FILE2)) OPTION(*ALL) QRYSLT('SOC +                                            
 7700                           *EQ "' *CAT &SOC *CAT '" & ETAB *EQ "' +                                               
 7800                           *CAT &ETAB *CAT '" &  %DIGITS(MCUM) *GE +                                             
 7900                           "' *CAT &MCUM *CAT '" & %DIGITS(MCUM) *LE +                                          
 8000                           "' *CAT &MCUF *CAT '" & QFPP *NE 0') +                                                
 8100                           KEYFLD((K1) (K2) (K3))                                                               
 8400              CALL       PGM(RPG2) PARM(&SOC &ETAB &DATE &DATF &INRE &USER)                                
 8500              CLOF       OPNID(FILE2)                                                                        
 8600              RCLRSC                                                                                       
 
12800              DLTOVR     FILE(*ALL)                                                                
12900              CHGJOB     RUNPTY(20) STSMSG(*USRPRF)                                               
13000              RMVLIBLE   LIB(&BIB)                                                                      
13100              MONMSG     MSGID(CPF2103 CPF2104 CPF2106 CPF2110 +                                                  
13200                           CPF2113 CPF2118 CPF2176 CPF2182 CPF9807) +                                             
13300                           EXEC(DO)                                                                               
13400              DSPJOBLOG  OUTPUT(*PRINT)                                                                           
13500              GOTO       CMDLBL(£ENDPGM)                                                                          
13600              ENDDO                                                                                               
13700              GOTO £ENDPGM                                                                                        
13800 /* -------------------------------------------------------------------------- */                                 
13900 /* TRAITEMENT DES ERREURS */                                                                                     
14000  ERREUR:                                                                                                         
14100 /* Réception des Messages */                                                                                     
14200  RCVMSG:     RCVMSG     MSGDTA(&MSGDTA) MSGID(&MSGID) +                                                          
14300                           RTNTYPE(&RTNTYPE) MSGF(&MSGF) +                                                        
14400                           MSGFLIB(&MSGFLIB)                                                                      
14500 /* Renvoi Messages */                                                                                            
14600              IF         COND(&MSGID *NE ' ') THEN(DO)                                                            
14700              IF         COND((&RTNTYPE = &DIAG) *OR (&RTNTYPE = +                                                
14800                           &EXCP)) THEN(DO)                                                                       
14900              SNDPGMMSG  MSGID(&MSGID) MSGF(&MSGFLIB/&MSGF) +                                                     
15000                           MSGDTA(&MSGDTA) TOPGMQ(*PRV) MSGTYPE(*DIAG)                                            
15100              MONMSG     MSGID(CPF0000)                                                                          
15200              ENDDO                                                                                               
15300              GOTO       CMDLBL(RCVMSG)                                                                           
15400              ENDDO                                                                                              
15500              ELSE       CMD(DO)                                                                                  
15600 /* Message de Fin anormale */                                                                                    
15700              SNDPGMMSG  MSGID(CPF9898) MSGF(QCPFMSG) +                                                           
15800                           MSGDTA(&CANCELD) TOPGMQ(*PRV) MSGTYPE(*DIAG)                                           
15900              MONMSG     MSGID(CPF0000)                                                                           
16000              ENDDO                                                                                               
16100 /* -------------------------------------------------------------------------- */                                 
16200 /* Fin de programme */                                                                                           
16300  £ENDPGM:                                                                                                        
16400              RCLRSC                                                                                              
16500              ENDPGM                                                                                              
 


